BLOCKSEC

Security Audit

Report for Puffer Fast Path Contracts

Date: September 3, 2024 **Version:** 1.0

Contact: contact@blocksec.com

Contents

Report Manifest

Item	Description
Client	Puffer Finance
Target	Puffer Fast Path Contracts

Version History

Version	Date	Description
1.0	September 3, 2024	First release

Signature

Digitally signed by: BlockSec (https://blocksec.com)

SHA1 digest of public key: BAB106D3708BC6DE0963CE05E9478C24B326AE1C

Date: 2024-09-04 09:33:51+0000

About BlockSec BlockSec focuses on the security of the blockchain ecosystem and collaborates with leading DeFi projects to secure their products. BlockSec is founded by top-notch security researchers and experienced experts from both academia and industry. They have published multiple blockchain security papers in prestigious conferences, reported several zero-day attacks of DeFi applications, and successfully protected digital assets that are worth more than 14 million dollars by blocking multiple attacks. They can be reached at Email, Twitter and Medium.

Chapter 1 Introduction

1.1 About Target Contracts

Information	Description
Type	Smart Contract
Language	Solidity
Approach	Semi-automatic and manual verification

The focus of this audit is on Puffer Fast Path Contracts [1] of Puffer Finance. The feature of Puffer Fast Path Contracts is designed to help the SNO (Staking Node Operators) to cheaply and quickly withdraw their rewards on the L2 network. The contracts covered in this audit include:

```
1 mainnet-contracts/script/DeployFWR.s.sol
2 mainnet-contracts/src/PufferVaultV3.sol
3 mainnet-contracts/src/L1RewardManager.sol
4 l2-contracts/src/L2RewardManager.sol
5 mainnet-contracts/src/RestakingOperator.sol
6 mainnet-contracts/src/PufferModuleManager.sol
7 mainnet-contracts/src/PufferModule.sol
```

Listing 1.1: Audit Scope for this Report

It is important to note that the scope does not cover full files for the `RestakingOperator`, `PufferModuleManager`, and `PufferModule` contracts. We merely focus on code updates of these contracts to support the EigenLayer M4 upgrade.

Other files are not within the scope of the audit. Additionally, all dependencies of the smart contracts within the audit scope are considered reliable in terms of both functionality and security, and are therefore not included in the audit scope.

The auditing process is iterative. Specifically, we would audit the commits that fix the discovered issues. If there are new issues, we will continue this process. The commit SHA values during the audit are shown in the following table. Our audit report is responsible for the code in the initial version (`Version 1`), as well as new code (in the following versions) to fix issues in the audit report.

Project	Version	Commit Hash
Puffer Fast Path Contracts	`Version 1`	`f058bb757636e754420a7d350ee561b087242ce2`
	`Version 2`	`1748779b57c2b8e14f7c860eae1b99cdbf8550b7`

1.2 Disclaimer

This audit report does not constitute investment advice or a personal recommendation. It does not consider, and should not be interpreted as considering or having any bearing on,

[1] https://github.com/PufferFinance/puffer-contracts

the potential economics of a token, token sale or any other product, service or other asset. Any entity should not rely on this report in any way, including for the purpose of making any decisions to buy or sell any token, product, service or other asset.

This audit report is not an endorsement of any particular project or team, and the report does not guarantee the security of any particular project. This audit does not give any warranties on discovering all security issues of the smart contracts, i.e., the evaluation result does not guarantee the nonexistence of any further findings of security issues. As one audit cannot be considered comprehensive, we always recommend proceeding with independent audits and a public bug bounty program to ensure the security of smart contracts.

The scope of this audit is limited to the code mentioned in Section 1.1. Unless explicitly specified, the security of the language itself (e.g., the solidity language), the underlying compiling toolchain and the computing infrastructure are out of the scope.

1.3 Procedure of Auditing

We perform the audit according to the following procedure.

- **Vulnerability Detection** We first scan smart contracts with automatic code analyzers, and then manually verify (reject or confirm) the issues reported by them.
- **Semantic Analysis** We study the business logic of smart contracts and conduct further investigation on the possible vulnerabilities using an automatic fuzzing tool (developed by our research team). We also manually analyze possible attack scenarios with independent auditors to cross-check the result.
- **Recommendation** We provide some useful advice to developers from the perspective of good programming practice, including gas optimization, code style, and etc.

We show the main concrete checkpoints in the following.

1.3.1 Software Security

* Reentrancy
* DoS
* Access control
* Data handling and data flow
* Exception handling
* Untrusted external call and control flow
* Initialization consistency
* Events operation
* Error-prone randomness
* Improper use of the proxy system

1.3.2 DeFi Security

* Semantic consistency
* Functionality consistency
* Permission management

- Business logic
- Token operation
- Emergency mechanism
- Oracle security
- Whitelist and blacklist
- Economic impact
- Batch transfer

1.3.3 NFT Security

- Duplicated item
- Verification of the token receiver
- Off-chain metadata security

1.3.4 Additional Recommendation

- Gas optimization
- Code quality and style

Note *The previous checkpoints are the main ones. We may use more checkpoints during the auditing process according to the functionality of the project.*

1.4 Security Model

To evaluate the risk, we follow the standards or suggestions that are widely adopted by both industry and academy, including OWASP Risk Rating Methodology [2] and Common Weakness Enumeration [3]. The overall *severity* of the risk is determined by *likelihood* and *impact*. Specifically, likelihood is used to estimate how likely a particular vulnerability can be uncovered and exploited by an attacker, while impact is used to measure the consequences of a successful exploit.

Table 1.1: Vulnerability Severity Classification

Impact \ Likelihood	*High*	*Low*
High	High	Medium
Low	Medium	Low

[2] https://owasp.org/www-community/OWASP_Risk_Rating_Methodology

[3] https://cwe.mitre.org/

In this report, both likelihood and impact are categorized into two ratings, i.e., *high* and *low* respectively, and their combinations are shown in Table 1.1.

Accordingly, the severity measured in this report are classified into three categories: **High**, **Medium**, **Low**. For the sake of completeness, **Undetermined** is also used to cover circumstances when the risk cannot be well determined.

Furthermore, the status of a discovered item will fall into one of the following four categories:

- **Undetermined** No response yet.
- **Acknowledged** The item has been received by the client, but not confirmed yet.
- **Confirmed** The item has been recognized by the client, but not fixed yet.
- **Fixed** The item has been confirmed and fixed by the client.

Chapter 2 Findings

In total, we found **eight** potential security issues. Besides, we have **four** recommendations and **one** note.

- Medium Risk: 3
- Low Risk: 5
- Recommendation: 4
- Note: 1

ID	Severity	Description	Category	Status
1	Low	Possible overflow in function `_handleMintAndBridge()`	Software Security	Fixed
2	Medium	Potential stuck of `xPufETH` when reverting mint rewards	DeFi Security	Fixed
3	Medium	Lack of check on freeze status in function `revertInterval()`	DeFi Security	Fixed
4	Medium	Incorrect checks in function `_setClaimingDelay()`	DeFi Security	Fixed
5	Low	Possible incorrect `destinationDomainId` for function `_revertInterval()`	DeFi Security	Confirmed
6	Low	Lack of checks on `allowedRewardMintFrequency`	DeFi Security	Fixed
7	Low	Users receive less `xPufETH` after the mint rewards reversion	DeFi Security	Confirmed
8	Low	Lack of check on origin domain in function `xReceive()`	DeFi Security	Fixed
9	-	Use returned identifier from function `createSelectFork()`	Recommendation	Fixed
10	-	Handle possibly dust `xPufETH` tokens in contract `L2RewardManager`	Recommendation	Confirmed
11	-	Ensure `epochRecord` exists in function `claimRewards()`	Recommendation	Fixed
12	-	Possible incorrect delegate in function `setL2RewardClaimer()`	Recommendation	Fixed
13	-	Potential centralization risks	Note	-

The details are provided in the following sections.

2.1 Software Security

2.1.1 Possible overflow in function `_handleMintAndBridge()`

Severity Low

Status Fixed in `Version 2`

Introduced by `Version 1`

Description The function `_handleMintAndBridge()` in the `L2RewardManager` contract contains

a potential overflow issue when dealing with the exchange rate exceeding `uint(64).max`. Although the likelihood of this extreme case may be minimal, it is advised to address such edge cases.

```
243   function _handleMintAndBridge(uint256 amount, bytes memory data) internal {
244       L1RewardManagerStorage.MintAndBridgeData memory params =
245           abi.decode(data, (L1RewardManagerStorage.MintAndBridgeData));
246
247       // Sanity check
248       if (amount != ((params.rewardsAmount * params.ethToPufETHRate) / 1 ether)) {
249           revert InvalidAmount();
250       }
251
252       RewardManagerStorage storage $ = _getRewardManagerStorage();
253
254       bytes32 intervalId = getIntervalId(params.startEpoch, params.endEpoch);
255
256       $.currentRewardsInterval = intervalId;
257
258       $.epochRecords[intervalId] = EpochRecord({
259           ethToPufETHRate: uint64(params.ethToPufETHRate),
260           startEpoch: uint72(params.startEpoch),
261           endEpoch: uint72(params.endEpoch),
262           timeBridged: uint48(block.timestamp),
263           rewardRoot: params.rewardsRoot,
264           pufETHAmount: uint128(amount),
265           ethAmount: uint128(params.rewardsAmount)
266       });
267
268       emit RewardRootAndRatePosted({
269           rewardsAmount: params.rewardsAmount,
270           ethToPufETHRate: params.ethToPufETHRate,
271           startEpoch: params.startEpoch,
272           intervalId: intervalId,
273           endEpoch: params.endEpoch,
274           rewardsRoot: params.rewardsRoot
275       });
276   }
```

Listing 2.1: l2-contracts/src/L2RewardManager.sol

Impact Potential incorrect `ethToPufETHRate` recorded on L2.

Suggestion Use type `uint256` for `EpochRecord.ethToPufETHRate`.

2.2 DeFi Security

2.2.1 Potential stuck of `xPufETH` when reverting mint rewards

Severity Medium

Status Fixed in `Version 2`

Introduced by `Version 1`

Description The `xReceive()` function in the `L1RewardManager` contract invokes the function `revertMintRewards()` in `PufferVaultV3` after receiving the `xPufETH` bridged from the `L2RewardManager`. The function `revertMintRewards()` will decrease the `totalRewardMintAmount` by the specified revert amount. However, if rewards have already been distributed through the `depositRewards()` function, the `totalRewardMintAmount` might be insufficient, leading to a revert due to an underflow issue.

Furthermore, according to the bridge's documentation [1], if the call on the receiver contract (also referred to as the 'target' contract) reverts, funds sent with the call will remain in the receiver contract. In this case, the `xPufETH` will get stuck in the `L1RewardManager` contract.

```
97  function xReceive(bytes32, uint256, address, address originSender, uint32, bytes memory
        callData)
98      external
99      override(IXReceiver)
100     restricted
101     returns (bytes memory)
102 {
103     // The call must originate from the L2_REWARDS_MANAGER
104     if (originSender != address(L2_REWARDS_MANAGER)) {
105         revert Unauthorized();
106     }
107
108     // We decode the data to get the amount of shares(pufETH) and the ETH amount.
109     L2RewardManagerStorage.EpochRecord memory epochRecord =
110         abi.decode(callData, (L2RewardManagerStorage.EpochRecord));
111
112     XPUFETH.approve(address(LOCKBOX), epochRecord.pufETHAmount);
113     // get the pufETH
114     LOCKBOX.withdraw(epochRecord.pufETHAmount);
115
116     // The PufferVault will burn the pufETH from this contract and subtract the ETH amount from
            the ethRewardsAmount
117     PUFFER_VAULT.revertMintRewards({ pufETHAmount: epochRecord.pufETHAmount, ethAmount:
            epochRecord.ethAmount });
118
119     emit RevertedRewards({
120         rewardsAmount: epochRecord.ethAmount,
121         startEpoch: epochRecord.startEpoch,
122         endEpoch: epochRecord.endEpoch,
123         rewardsRoot: epochRecord.rewardRoot
124     });
125
126     return "";
127 }
```

Listing 2.2: mainnet-contracts/src/L1RewardManager.sol

```
97  function depositRewards() external payable restricted {
98      VaultStorage storage $ = _getPufferVaultStorage();
99      uint256 previousRewardsAmount = $.totalRewardMintAmount;
```

[1] https://docs.connext.network/developers/guides/handling-failures

```
100         uint256 newTotalRewardsAmount = previousRewardsAmount - msg.value;
101         $.totalRewardMintAmount = newTotalRewardsAmount;
102
103         emit UpdatedTotalRewardsAmount(previousRewardsAmount, newTotalRewardsAmount, msg.value);
104     }
105
106     /**
107      * @notice Reverts the `mintRewards` action.
108      * @dev Restricted to L1RewardManager
109      */
110     function revertMintRewards(uint256 pufETHAmount, uint256 ethAmount) external restricted {
111         VaultStorage storage $ = _getPufferVaultStorage();
112
113         uint256 previousRewardsAmount = $.totalRewardMintAmount;
114         uint256 newTotalRewardsAmount = previousRewardsAmount - ethAmount;
115         $.totalRewardMintAmount = newTotalRewardsAmount;
116
117         emit UpdatedTotalRewardsAmount(previousRewardsAmount, newTotalRewardsAmount, 0);
118
119         // msg.sender is the L1RewardManager contract
120         _burn(msg.sender, pufETHAmount);
121     }
```

Listing 2.3: mainnet-contracts/src/PufferVaultV3.sol

Impact Bridged `xPufETH` tokens could get stuck in the `L1RewardManager` contract, resulting in an inconsistent state and causing some users to lose their rewards.

Suggestion Revise the function `depositRewards()` to maintain a separate variable that tracks already distributed rewards.

2.2.2 Lack of check on freeze status in function `revertInterval()`

Severity Medium

Status Fixed in `Version 2`

Introduced by `Version 1`

Description The function `revertInterval()` in the `L2RewardManager` contract does not check the interval's freeze status. This is inconsistent with the code annotation on line 165, which suggests that only a frozen interval can be reverted. Moreover, while the function `_freezeClaimingForInterval()` forbids reverting an unlocked interval (lines 328-331), `revertInterval()` omits such sanity checks. Consequently, an unlocked interval can also be reverted, leading to inconsistencies between L1 and L2 accounting.

```
164     /**
165      * @notice Reverts the already frozen interval. It bridges the xPufETH back to the L1
166      * @dev On the L1, we unwrap xPufETH to pufETH and burn the pufETH to undo the minting
167      * We use msg.value to pay for the relayer fee on the destination chain.
168      */
169     function revertInterval(address bridge, uint256 startEpoch, uint256 endEpoch) external payable
            restricted {
170         _revertInterval(bridge, startEpoch, endEpoch);
```

```
171 }
```

Listing 2.4: l2-contracts/src/L2RewardManager.sol

```
323 function _freezeClaimingForInterval(uint256 startEpoch, uint256 endEpoch) internal {
324     RewardManagerStorage storage $ = _getRewardManagerStorage();
325
326     bytes32 intervalId = getIntervalId(startEpoch, endEpoch);
327
328     // Revert if the claiming is not locked
329     if (!_isClaimingLocked(intervalId)) {
330         revert UnableToFreezeInterval();
331     }
332
333     // revert for non-existing interval
334     if ($.epochRecords[intervalId].rewardRoot == bytes32(0)) {
335         revert UnableToFreezeInterval();
336     }
337
338     // To freeze the claiming, we set the timeBridged to 0
339     $.epochRecords[intervalId].timeBridged = 0;
340
341     emit ClaimingIntervalFrozen({ startEpoch: startEpoch, endEpoch: endEpoch });
342 }
```

Listing 2.5: l2-contracts/src/L2RewardManager.sol

```
347 function _revertInterval(address bridge, uint256 startEpoch, uint256 endEpoch) internal {
348     RewardManagerStorage storage $ = _getRewardManagerStorage();
349
350     BridgeData memory bridgeData = $.bridges[bridge];
351
352     if (bridgeData.destinationDomainId == 0) {
353         revert BridgeNotAllowlisted();
354     }
355
356     bytes32 intervalId = getIntervalId(startEpoch, endEpoch);
357
358     EpochRecord memory epochRecord = $.epochRecords[intervalId];
359
360     XPUFETH.approve(bridge, epochRecord.pufETHAmount);
361
362     IBridgeInterface(bridge).xcall{ value: msg.value }({
363         destination: bridgeData.destinationDomainId, // Domain ID of the destination chain
364         to: L1_REWARD_MANAGER, // Address of the target contract
365         asset: address(XPUFETH), // Address of the token contract
366         delegate: msg.sender, // Address that can revert or forceLocal on destination
367         amount: epochRecord.pufETHAmount, // Amount of tokens to transfer
368         slippage: 0, // Max slippage the user will accept in BPS (e.g. 300 = 3%)
369         callData: abi.encode(epochRecord) // Encoded data to send
370      });
371
372     delete $.epochRecords[intervalId];
```

```
373
374         emit ClaimingIntervalReverted({
375             startEpoch: startEpoch,
376             endEpoch: endEpoch,
377             intervalId: intervalId,
378             pufETHAmount: epochRecord.pufETHAmount,
379             rewardsRoot: epochRecord.rewardRoot
380         });
381     }
```

Listing 2.6: l2-contracts/src/L2RewardManager.sol

Impact Reverts on unfrozen intervals could potentially disrupt the accounting and reward functionality.

Suggestion Add a check to ensure the interval is frozen (i.e., `timeBridged == 0 && rewardRoot != bytes32(0)`) in the function `revertInterval()`.

2.2.3 Incorrect checks in function `_setClaimingDelay()`

Severity Medium

Status Fixed in `Version 2`

Introduced by `Version 1`

Description The function `_setClaimingDelay()` will revert if the current claiming interval is not locked and the new delayed timestamp (i.e., `timeBridged + newDelay`) exceeds the current timestamp. However, when the current claiming interval is locked, the `newDelay` can be set to any value more than 6 hours, which may make older unlocked claiming intervals being locked again.

```
291     function _setClaimingDelay(uint256 newDelay) internal {
292         if (newDelay < 6 hours) {
293             revert InvalidDelayPeriod();
294         }
295         RewardManagerStorage storage $ = _getRewardManagerStorage();
296
297         // Revert only if the claiming is not locked and the new delayed timestamp (timeBridged+
                newDelay) exceedes the current timestamp
298         if (
299             !_isClaimingLocked($.currentRewardsInterval)
300                 && ($.epochRecords[$.currentRewardsInterval].timeBridged + newDelay > block.
                        timestamp)
301         ) {
302             revert RelockingIntervalIsNotAllowed();
303         }
304
305         emit ClaimingDelayChanged({ oldDelay: $.claimingDelay, newDelay: newDelay });
306         $.claimingDelay = newDelay;
307     }
```

Listing 2.7: l2-contracts/src/L2RewardManager.sol

Impact Unlocked claiming intervals can be changed to locked.

Suggestion Track every interval's unlocked time using a separate data struct (e.g., mapping) and invoking function `_setClaimingDelay()` will only affect current and future claiming intervals.

2.2.4 Possible incorrect `destinationDomainId` for function `_revertInterval()`

Severity Low

Status Confirmed

Introduced by `Version 1`

Description In the current implementation, the `_revertInterval` function in the `L2RewardManager` contract initiates a cross-chain call using the `bridgeData.destinationDomainId` parameter. However, the function `updateBridgeData()` allows the `destinationDomainId` for a bridge to be updated, which could result in the cross-chain revert call being accidentally bridged to an incorrect target chain instead of the original one.

```
178   function updateBridgeData(address bridge, BridgeData memory bridgeData) external restricted {
179       RewardManagerStorage storage $ = _getRewardManagerStorage();
180
181       if (bridge == address(0)) {
182           revert InvalidAddress();
183       }
184
185       $.bridges[bridge].destinationDomainId = bridgeData.destinationDomainId;
186       emit BridgeDataUpdated(bridge, bridgeData);
187   }
```

Listing 2.8: l2-contracts/src/L2RewardManager.sol

```
347   function _revertInterval(address bridge, uint256 startEpoch, uint256 endEpoch) internal {
348       RewardManagerStorage storage $ = _getRewardManagerStorage();
349
350       BridgeData memory bridgeData = $.bridges[bridge];
351
352       if (bridgeData.destinationDomainId == 0) {
353           revert BridgeNotAllowlisted();
354       }
355
356       bytes32 intervalId = getIntervalId(startEpoch, endEpoch);
357
358       EpochRecord memory epochRecord = $.epochRecords[intervalId];
359
360       XPUFETH.approve(bridge, epochRecord.pufETHAmount);
361
362       IBridgeInterface(bridge).xcall{ value: msg.value }({
363           destination: bridgeData.destinationDomainId, // Domain ID of the destination chain
364           to: L1_REWARD_MANAGER, // Address of the target contract
365           asset: address(XPUFETH), // Address of the token contract
366           delegate: msg.sender, // Address that can revert or forceLocal on destination
367           amount: epochRecord.pufETHAmount, // Amount of tokens to transfer
```

```
368             slippage: 0, // Max slippage the user will accept in BPS (e.g. 300 = 3%)
369             callData: abi.encode(epochRecord) // Encoded data to send
370         });
371
372         delete $.epochRecords[intervalId];
373
374         emit ClaimingIntervalReverted({
375             startEpoch: startEpoch,
376             endEpoch: endEpoch,
377             intervalId: intervalId,
378             pufETHAmount: epochRecord.pufETHAmount,
379             rewardsRoot: epochRecord.rewardRoot
380         });
381     }
```

Listing 2.9: l2-contracts/src/L2RewardManager.sol

Impact The cross-chain message could be bridged to the incorrect chain, leading to unexpected results.

Suggestion Record the original `destinationDomainId` in the `$.epochRecords` for each interval.

Feedback from the project The domainId doesn't change (Connext is using a constant for it). We added this `updateBridgeData()` function to be able to swap out `Connext` (in case some other bridge decides to use the same interface), although this is an unlikely scenario.

2.2.5 Lack of checks on `allowedRewardMintFrequency`

Severity Low

Status Fixed in `Version 2`

Introduced by `Version 1`

Description Currently, the function `mintAndBridgeRewards()` doesn't check whether `allowedRewardMintFrequency` is zero, which is not initialized in the function `initialize()`. If function `mintAndBridgeRewards()` is invoked before `allowedRewardMintFrequency` is initialized, the mint frequency check will be bypassed and lead to unexpected results.

Additionally, the function `setAllowedRewardMintFrequency()` doesn't ensure that the new frequency is bigger than a minimal value, which may bring possible risks.

```
99      function mintAndBridgeRewards(MintAndBridgeParams calldata params) external payable restricted
            {
100         RewardManagerStorage storage $ = _getRewardManagerStorage();
101
102         if (params.rewardsAmount > $.allowedRewardMintAmount) {
103             revert InvalidMintAmount();
104         }
105
106         if (($.lastRewardMintTimestamp + $.allowedRewardMintFrequency) > block.timestamp) {
107             revert NotAllowedMintFrequency();
108         }
```

Listing 2.10: mainnet-contracts/src/L1RewardManager.sol

```
229   function setAllowedRewardMintFrequency(uint104 newFrequency) external restricted {
230       RewardManagerStorage storage $ = _getRewardManagerStorage();
231
232       emit AllowedRewardMintFrequencyUpdated($.allowedRewardMintFrequency, newFrequency);
233
234       $.allowedRewardMintFrequency = newFrequency;
235   }
```

Listing 2.11: src/rewards/MultiRewardDistributor.sol

Impact The frequency check in function `mintAndBridgeRewards()` can be ineffective.

Suggestion Add checks on `allowedRewardMintFrequency` in functions `mintAndBridgeRewards()` and `setAllowedRewardMintFrequency()` accordingly.

2.2.6 Users receive less `xPufETH` after the mint rewards reversion

Severity Low

Status Confirmed

Introduced by `Version 1`

Description The `pufETH` token is an interest-bearing token that is expected to increase in value as `PufferVault` receives validator rewards. The function `mintRewards()` in the `PufferVaultV3` contract calculates the `pufETHAmount` using the spot price returned by `convertToShares(1 ether)`. The function `revertMintRewards()` burns a corresponding `pufETHAmount` of `pufETH` tokens, based on the exchange rate at the time of minting. Due to the potential increase in `pufETH`'s price between the minting and reverting periods, users may lose a portion of their rewards when the rewards are re-minted and bridged.

```
70   function mintRewards(uint256 rewardsAmount)
71       external
72       restricted
73       returns (uint256 ethToPufETHRate, uint256 pufETHAmount)
74   {
75       ethToPufETHRate = convertToShares(1 ether);
76       // calculate the shares using this formula since calling convertToShares again is costly
77       pufETHAmount = ethToPufETHRate.mulDiv(rewardsAmount, 1 ether, Math.Rounding.Floor);
78
79       VaultStorage storage $ = _getPufferVaultStorage();
80
81       uint256 previousRewardsAmount = $.totalRewardMintAmount;
82       uint256 newTotalRewardsAmount = previousRewardsAmount + rewardsAmount;
83       $.totalRewardMintAmount = newTotalRewardsAmount;
84
85       emit UpdatedTotalRewardsAmount(previousRewardsAmount, newTotalRewardsAmount, 0);
86
87       // msg.sender is the L1RewardManager contract
88       _mint(msg.sender, pufETHAmount);
89
90       return (ethToPufETHRate, pufETHAmount);
91   }
```

Listing 2.12: mainnet-contracts/src/PufferVaultV3.sol

```
163   function xReceive(bytes32, uint256, address, address originSender, uint32, bytes memory
          callData)
164       external
165       override(IXReceiver)
166       restricted
167       returns (bytes memory)
168   {
169       // The call must originate from the L2_REWARDS_MANAGER
170       if (originSender != address(L2_REWARDS_MANAGER)) {
171           revert Unauthorized();
172       }
173
174       // We decode the data to get the amount of shares(pufETH) and the ETH amount.
175       L2RewardManagerStorage.EpochRecord memory epochRecord =
176           abi.decode(callData, (L2RewardManagerStorage.EpochRecord));
177
178       XPUFETH.approve(address(LOCKBOX), epochRecord.pufETHAmount);
179       // get the pufETH
180       LOCKBOX.withdraw(epochRecord.pufETHAmount);
181
182       // The PufferVault will burn the pufETH from this contract and subtract the ETH amount from
              the ethRewardsAmount
183       PUFFER_VAULT.revertMintRewards({ pufETHAmount: epochRecord.pufETHAmount, ethAmount:
              epochRecord.ethAmount });
184
185       emit RevertedRewards({
186           rewardsAmount: epochRecord.ethAmount,
187           startEpoch: epochRecord.startEpoch,
188           endEpoch: epochRecord.endEpoch,
189           rewardsRoot: epochRecord.rewardRoot
190       });
191
192       return "";
193   }
```

Listing 2.13: mainnet-contracts/src/L1RewardManager.sol

```
110   function revertMintRewards(uint256 pufETHAmount, uint256 ethAmount) external restricted {
111       VaultStorage storage $ = _getPufferVaultStorage();
112
113       uint256 previousRewardsAmount = $.totalRewardMintAmount;
114       uint256 newTotalRewardsAmount = previousRewardsAmount - ethAmount;
115       $.totalRewardMintAmount = newTotalRewardsAmount;
116
117       emit UpdatedTotalRewardsAmount(previousRewardsAmount, newTotalRewardsAmount, 0);
118
119       // msg.sender is the L1RewardManager contract
120       _burn(msg.sender, pufETHAmount);
121   }
```

Listing 2.14: mainnet-contracts/src/PufferVaultV3.sol

Impact Staking node operators may receive less `xPufETH` when rewards are redistributed.

Suggestion Revise the logic to ensure the correct distribution of rewards for each interval.

Feedback from the project We acknowledge this issue and we will find a way to reimburse the node operators if this scenario happens. Coding the logic in the smart contract, would add complexity which we do not want.

2.2.7 Lack of check on origin domain in function `xReceive()`

Severity Low

Status Fixed in `Version 2`

Introduced by `Version 1`

Description In both `L1RewardManager` and `L2RewardManager` contracts, the function `xReceive()` should verify the validity of the origin data, as Connext's documentation [2] suggests. However, the current implementation lacks validation on the `_origin`, i.e., the origin domain ID.

```
163   function xReceive(bytes32, uint256, address, address originSender, uint32, bytes memory
          callData)
164       external
165       override(IXReceiver)
166       restricted
167       returns (bytes memory)
168   {
169       // The call must originate from the L2_REWARDS_MANAGER
170       if (originSender != address(L2_REWARDS_MANAGER)) {
171           revert Unauthorized();
172       }
173
174       // We decode the data to get the amount of shares(pufETH) and the ETH amount.
175       L2RewardManagerStorage.EpochRecord memory epochRecord =
176           abi.decode(callData, (L2RewardManagerStorage.EpochRecord));
177
178       XPUFETH.approve(address(LOCKBOX), epochRecord.pufETHAmount);
179       // get the pufETH
180       LOCKBOX.withdraw(epochRecord.pufETHAmount);
181
182       // The PufferVault will burn the pufETH from this contract and subtract the ETH amount from
              the ethRewardsAmount
183       PUFFER_VAULT.revertMintRewards({ pufETHAmount: epochRecord.pufETHAmount, ethAmount:
              epochRecord.ethAmount });
184
185       emit RevertedRewards({
186           rewardsAmount: epochRecord.ethAmount,
187           startEpoch: epochRecord.startEpoch,
188           endEpoch: epochRecord.endEpoch,
```

[2] https://docs.connext.network/developers/guides/authentication

```
189             rewardsRoot: epochRecord.rewardRoot
190         });
191
192         return "";
193     }
```

Listing 2.15: mainnet-contracts/src/L1RewardManager.sol

```
113     function xReceive(bytes32, uint256 amount, address, address originSender, uint32, bytes memory
            callData)
114         external
115         override(IXReceiver)
116         restricted
117         returns (bytes memory)
118     {
119         if (originSender != address(L1_REWARD_MANAGER)) {
120             revert Unauthorized();
121         }
122
123         IL1RewardManager.BridgingParams memory bridgingParams = abi.decode(callData, (
                IL1RewardManager.BridgingParams));
124
125         if (bridgingParams.bridgingType == IL1RewardManager.BridgingType.MintAndBridge) {
126             _handleMintAndBridge(amount, bridgingParams.data);
127         } else if (bridgingParams.bridgingType == IL1RewardManager.BridgingType.SetClaimer) {
128             _handleSetClaimer(bridgingParams.data);
129         }
130         // Return empty data
131         return "";
132     }
```

Listing 2.16: mainnet-contracts/src/L1RewardManager.sol

Impact Executing calls from unexpected origin domains may disrupt the functionality.

Suggestion Add a whitelist to hold supported domains and check `_origin` against the whitelist.

2.3 Additional Recommendation

2.3.1 Use returned identifier from function `createSelectFork()`

Status Fixed in `Version 2`

Introduced by `Version 1`

Description It's recommended to use the returned identifier from the function `createSelectFork()` as the parameter for the function `selectFork()`, rather than use a hardcoded zero value in contract `DeployFWR`.

```
25     function run() public {
26         GenerateAccessManagerCalldata3 generator = new GenerateAccessManagerCalldata3();
27
```

```
28        vm.createSelectFork(vm.rpcUrl("mainnet"));
29
30        vm.startBroadcast();
31
32        address noImpl = address(new NoImplementation());
33
34        // Deploy empty proxy
35        l1RewardManagerProxy = address(new ERC1967Proxy(noImpl, ""));
36
37        vm.label(address(l1RewardManagerProxy), "l1RewardManagerProxy");
38
39        // Generate L1 calldata
40        bytes memory l1AccessManagerCalldata = generator.generateL1Calldata({
41            l1RewardManagerProxy: l1RewardManagerProxy,
42            l1Bridge: _getEverclear(),
43            pufferVaultProxy: _getPufferVault(),
44            pufferModuleManagerProxy: _getPufferModuleManager()
45        });
46
47        console.log("L1 Access Manager Calldata");
48        console.logBytes(l1AccessManagerCalldata);
49
50        vm.stopBroadcast();
51
52        // Deploy contracts on L2
53        vm.createSelectFork(vm.rpcUrl("base"));
54        vm.startBroadcast();
55
56        L2RewardManager newImplementation = new L2RewardManager(_getXPufETH(), address(
              l1RewardManagerProxy));
57
58        console.log("L2RewardManager Implementation", address(newImplementation));
59
60        l2RewardManagerProxy = address(
61            new ERC1967Proxy(
62                address(newImplementation), abi.encodeCall(L2RewardManager.initialize, (
                      _getAccessManager()))
63            )
64        );
65        vm.makePersistent(l2RewardManagerProxy);
66
67        console.log("L2RewardManager Proxy", address(l2RewardManagerProxy));
68        vm.label(address(l2RewardManagerProxy), "L2RewardManagerProxy");
69        vm.label(address(newImplementation), "L2RewardManagerImplementation");
70
71        bytes memory l2AccessManagerCalldata =
72            generator.generateL2Calldata({ l2RewardManagerProxy: l2RewardManagerProxy, l2Bridge:
                  _getEverclear() });
73
74        console.log("L2 Access Manager Calldata");
75        console.logBytes(l2AccessManagerCalldata);
76
77        // Upgrade contract on L1
```

```
78        vm.stopBroadcast();
79
80        // Switch back to Fork 0
81        vm.selectFork(0);
82        vm.startBroadcast();
```

Listing 2.17: mainnet-contracts/script/DeployFWR.s.sol

Suggestion Use the returned identifier from the function `createSelectFork()` for function `selectFork()`.

2.3.2 Handle possibly dust `xPufETH` tokens in contract `L2RewardManager`

Status Confirmed

Introduced by `Version 1`

Description In the `L1RewardManager` contract, the function `mintAndBridge()` calculates the `pufETHAmount` on Line 77 based on the total rewards for all validators (i.e., `ethToPufETHRate-.mulDiv(rewardsAmount, 1 ether, Math.Rounding.Floor)`). However, when users claim rewards on L2, each user's reward amount is calculated based on their individual share in the `claimRewards` function. The cumulative precision loss for each user can result in residual dust `xPufETH` tokens in the `L2RewardManager` contract.

```
70    function mintRewards(uint256 rewardsAmount)
71        external
72        restricted
73        returns (uint256 ethToPufETHRate, uint256 pufETHAmount)
74    {
75        ethToPufETHRate = convertToShares(1 ether);
76        // calculate the shares using this formula since calling convertToShares again is costly
77        pufETHAmount = ethToPufETHRate.mulDiv(rewardsAmount, 1 ether, Math.Rounding.Floor);
78
79        VaultStorage storage $ = _getPufferVaultStorage();
80
81        uint256 previousRewardsAmount = $.totalRewardMintAmount;
82        uint256 newTotalRewardsAmount = previousRewardsAmount + rewardsAmount;
83        $.totalRewardMintAmount = newTotalRewardsAmount;
84
85        emit UpdatedTotalRewardsAmount(previousRewardsAmount, newTotalRewardsAmount, 0);
86
87        // msg.sender is the L1RewardManager contract
88        _mint(msg.sender, pufETHAmount);
89
90        return (ethToPufETHRate, pufETHAmount);
91    }
```

Listing 2.18: mainnet-contracts/src/PufferVaultV3.sol

```
55    function claimRewards(ClaimOrder[] calldata claimOrders) external restricted {
56        for (uint256 i = 0; i < claimOrders.length; i++) {
57            if (isClaimed(claimOrders[i].intervalId, claimOrders[i].account)) {
58                revert AlreadyClaimed(claimOrders[i].intervalId, claimOrders[i].account);
```

```
59          }
60
61          RewardManagerStorage storage $ = _getRewardManagerStorage();
62
63          // L1 contracts MUST set the claimer
64          address recipient = $.rewardsClaimers[claimOrders[i].account];
65          if (claimOrders[i].isL1Contract && recipient == address(0)) {
66              revert ClaimerNotSet(claimOrders[i].account);
67          }
68
69          EpochRecord storage epochRecord = $.epochRecords[claimOrders[i].intervalId];
70
71          if (_isClaimingLocked(claimOrders[i].intervalId)) {
72              revert ClaimingLocked({
73                  intervalId: claimOrders[i].intervalId,
74                  account: claimOrders[i].account,
75                  lockedUntil: epochRecord.timeBridged + $.claimingDelay
76              });
77          }
78
79          // Alice may run many Puffer validators in the same interval `totalETHEarned = sum(
                aliceValidators)`
80          // The leaf is: keccak256(abi.encode(AliceAddress, isL1Contract, totalETHEarned))
81          bytes32 leaf = keccak256(
82              bytes.concat(
83                  keccak256(abi.encode(claimOrders[i].account, claimOrders[i].isL1Contract,
                        claimOrders[i].amount))
84              )
85          );
86          if (!MerkleProof.verifyCalldata(claimOrders[i].merkleProof, epochRecord.rewardRoot,
                leaf)) {
87              revert InvalidProof();
88          }
89
90          // Mark it claimed and transfer the tokens
91          $.claimedRewards[claimOrders[i].intervalId][claimOrders[i].account] = true;
92
93          uint256 amountToTransfer = (claimOrders[i].amount * epochRecord.ethToPufETHRate) / 1
                ether;
94
95          recipient = recipient == address(0) ? claimOrders[i].account : recipient;
96
97          // if the custom claimer is set, then transfer the tokens to the set claimer
98          XPUFETH.transfer(recipient, amountToTransfer);
99
100         emit Claimed({
101             recipient: recipient,
102             account: claimOrders[i].account,
103             intervalId: claimOrders[i].intervalId,
104             amount: amountToTransfer
105         });
106     }
107 }
```

Listing 2.19: l2-contracts/src/L2RewardManager.sol

Suggestion Add a function to handle the dust `xPufETH` left in the `L2RewardManager` contract.

Feedback from the project Acknowledged. The contract is upgradeable, if the dust gets big enough, we will handle it in the future.

2.3.3 Ensure `epochRecord` exists in function `claimRewards()`

Status Fixed in `Version 2`

Introduced by `Version 1`

Description Function `claimRewards()` doesn't check if `epochRecord` exists before invoking function `_isClaimingLocked()`, which may trigger a `ClaimingLocked` revert with wrong information.

```
55    function claimRewards(ClaimOrder[] calldata claimOrders) external restricted {
56        for (uint256 i = 0; i < claimOrders.length; i++) {
57            if (isClaimed(claimOrders[i].intervalId, claimOrders[i].account)) {
58                revert AlreadyClaimed(claimOrders[i].intervalId, claimOrders[i].account);
59            }
60
61            RewardManagerStorage storage $ = _getRewardManagerStorage();
62
63            // L1 contracts MUST set the claimer
64            address recipient = $.rewardsClaimers[claimOrders[i].account];
65            if (claimOrders[i].isL1Contract && recipient == address(0)) {
66                revert ClaimerNotSet(claimOrders[i].account);
67            }
68
69            EpochRecord storage epochRecord = $.epochRecords[claimOrders[i].intervalId];
70
71            if (_isClaimingLocked(claimOrders[i].intervalId)) {
72                revert ClaimingLocked({
73                    intervalId: claimOrders[i].intervalId,
74                    account: claimOrders[i].account,
75                    lockedUntil: epochRecord.timeBridged + $.claimingDelay
76                });
77            }
```

Listing 2.20: l2-contracts/src/L2RewardManager.sol

Suggestion Check if the `epochRecord` exists and revert with a specific error.

2.3.4 Possible incorrect delegate in function `setL2RewardClaimer()`

Status Fixed in `Version 2`

Introduced by `Version 1`

Description The `delegate` parameter of the bridge call is set as the `msg.sender` in function `setL2RewardClaimer()`. However, the user (i.e., `msg.sender`) on L1 may not control the same address on L2, which may bring risks since the address of the delegate can update slippage and other bridge parameters on the destination chain.

```
61  function setL2RewardClaimer(address bridge, address claimer) external payable {
62      RewardManagerStorage storage $ = _getRewardManagerStorage();
63
64      BridgeData memory bridgeData = $.bridges[bridge];
65
66      if (bridgeData.destinationDomainId == 0) {
67          revert BridgeNotAllowlisted();
68      }
69
70      // msg.value is used to pay for the relayer fee on the destination chain
71      IBridgeInterface(bridge).xcall{ value: msg.value }({
72          destination: bridgeData.destinationDomainId, // Domain ID of the destination chain
73          to: L2_REWARDS_MANAGER, // Address of the target contract on the destination chain
74          delegate: msg.sender, // Address that can revert or forceLocal on destination
75          asset: address(0), // Address of the token contract
76          amount: 0, // We don't transfer any tokens
77          slippage: 0, // No slippage
78          callData: abi.encode(
79              BridgingParams({
80                  bridgingType: BridgingType.SetClaimer,
81                  data: abi.encode(SetClaimerParams({ account: msg.sender, claimer: claimer }))
82              })
83          ) // Encoded data to bridge to the target contract
84       });
```

Listing 2.21: mainnet-contracts/src/L1RewardManager.sol

Suggestion Let the user pass the delegate as a parameter of the function `setL2RewardClaimer()`, or use an address controlled by the puffer protocol.

2.4 Note

2.4.1 Potential centralization risks

Introduced by `Version 1`

Description In `puffer-contracts`, there are some privileged functions to update critical configurations, such as the bridge and mint configurations. These functions have restricted modifiers and are claimed to be controlled by a multi-signature wallet. If most of the private keys in this wallet are controlled by a single entity, or if the private keys are leaked. The protocol can be potentially incapacitated.


BOOST WEB3 THROUGH NEXT-GENERATION SECURITY & USABILITY INNOVATIONS